
;;SIM-HMCM(CGCBCA)CO - Tugarinov, Venditti & Clore 
 
#include <Avance.incl> 
#include <Grad.incl> 
 
#define CO 
#define METHYL 
 
define delay TAUB 
define delay TAUBS 
define delay TAU 
define delay DEUT 
define delay EPSI 
define delay TAUD 
define delay TAUD_1 
define delay TAUC 
define delay TAUC_1 
define delay TAUCS 
define delay pd15 
 
define pulse C_REB 
define pulse CA_REB 
define pulse VME_REB 
define pulse CO_REB 
define pulse CG2_REB 
define pulse Q5P 
 
"C_REB=p5"      ; high power RE-BURP (~350u on-resonance at 600 MHz) 
"CO_REB=p6" ; high power RE-BURP (~350u at 600 MHz applied at 176 ppm) 
"CG2_REB=p15"   ; 4.0 ms RE-BURP applied at 16 ppm for Ile Cg2 methyls (600 MHz) 
"CA_REB=p19"    ; 1.8 ms RE-BURP applied at 60 ppm for ILV Ca (600 MHz) 
"VME_REB=p9"    ; 3.0 ms RE-BURP applied at 21.5 ppm for Val Cg methyls (600 
MHz) 
"Q5P=p14"       ; 1.4 ms Q5 Gaussian Cascade applied on-resonance (33 ppm) 
 
"pd15=p15" 
"DEUT=1.4m" 
 
 
"TAUB=3.5m-8u" 
"TAUBS=TAUB-C_REB*0.5" 
"TAU=3.5m-DEUT-p17-8u" 
 
"TAUC=3.5m-8u-0.5*CG2_REB" 
"TAUC_1=TAUC" 

"TAUCS=3.5m-8u-0.5*CG2_REB" 
 
"TAUD=2.50m-4u-CO_REB*0.5-VME_REB*0.5" 
"TAUD_1=TAUD" 
 
"d3=2.0m-p25-12u" 
"d4=2.0m-p25-14u-p18-4u" ;  use for watergate (p18@pl9) in water 
;"d4=2.0m-p25-14u-4u" 
 
"d11=5m" 
"d12=5m" 
 
; carbonyl evolution delays 
"d0=2u" 
 
"EPSI=p21*0.5-p10" 
 
; methyl evolution delays 
"d22=3.5m - 4u" 
"d23=3.5m - 4u - C_REB*0.5" 
"d24=3.5m - 4u - C_REB*0.5" 
"d25=3.5m - 4u" 
 
;----------------------------------------------------------------------- 
"cnst21=2264"      ; Ca-Cb offset (33 ppm) 
"cnst22=0"    ; offset of 13C carrier (18 ppm; Methyl) 
;----------------------------------------------------------------------- 
 
1 ze 
d11 LOCKDEC_ON 
d11 H2_PULSE 
2 10u 
  d11 do:f2 
  d12 
3 2u 
4 2u 
5 2u 
6 2u 
7 2u 
8 2u 
  d11 H2_LOCK 
  d11 LOCKH_OFF 
  d1 
  d12 UNBLKGRAD 
  d12 H2_PULSE 
  50u fq=cnst22:f2 

  10u pl8:f2 
  10u pl0:f1 
  10u pl7:f3 
  10u pl17:f4 
  (p2:sp2 ph0:r):f1       ; 7-ms EBURP-1 on water 
  2u pl0:f1 
;;  (p3:sp3 ph0:r):f1 
  2u 
  (p8 ph0):f2 
  2u 
  p26:gp0 
  1m pl1:f1 
 
 
  (p1 ph0):f1 
  2u 
  p25:gp0 
  d3 
  10u 
  (center (p8*2 ph0):f2 (p1*2 ph0):f1) 
  2u 
  p25:gp0 
  d3 
  10u fq=cnst21:f2 
 
  (p8 ph2):f2 
  4u 
  TAUB 
  4u 
  (p1*2 ph0):f1 
  4u 
  TAUBS 
  4u pl0:f2 
  (C_REB:sp5 ph0):f2 
  4u 
  TAUBS 
  4u 
  (p1*2 ph0):f1 
  2u 
  2u pl8:f2 
  TAUB 
  2u pl18:f4 
  2u cpds4:f4 
  (lalign (p8 ph3):f2 (p1 ph11):f1) 
 
  4u 

  TAUC 
  4u pl0:f2 
  pd15   ;(CG1_REB:sp15 ph0):f2 
  4u 
  TAUCS 
  4u pl0:f2 
  (C_REB:sp5 ph0):f2 
  4u 
  TAUCS 
  4u pl0:f2 
  pd15   ;(CG1_REB:sp15 ph0):f2 
  4u 
  TAUC_1 
  4u 
   
  (Q5P:sp14 ph14):f2 
  4u 
  TAUC 
  4u pl0:f2 
  (CG2_REB:sp15 ph0):f2 
  4u 
  TAUCS 
  4u pl0:f2 
  (C_REB:sp5 ph0):f2                ; CHANGE THIS PHASE TO Y INVERT VAL VS. 
ILE/LEU 
  4u 
  TAUCS 
  4u pl0:f2 
  (CG2_REB:sp15 ph0):f2 
  4u 
  TAUC 
  4u pl8:f2 
  (p8 ph3):f2 
 
; ******* CARBON CA-CO TRANSFER ****** 
    TAUD pl0:f2 
    (VME_REB:sp9 ph0):f2 
    TAUD pl0:f2  
    2u pl0:f2 
    (CO_REB:sp6 ph0):f2 
    2u pl0:f2 
    (C_REB:sp5 ph0):f2 
    2u pl0:f2 
    TAUD_1 pl0:f2 
    (VME_REB:sp9 ph0):f2 
    TAUD_1 pl0:f2 

    (CO_REB:sp6 ph0):f2 
    2u pl0:f2 
    (p10:sp10 ph1):f2 
; ****** CARBON CA-CO TRANSFER END ***** 
 
    2u pl0:f2 
 
; ******** CARBONYL EVOLUTION ********** 
   (p10:sp11 ph5):f2   
   d0 
   2u pl0:f2 
   (center (CA_REB:sp19 ph15):f2 (p7*2 ph0):f3) 
   2u pl0:f2 
   d0 
   2u 
   (p10*2:sp13 ph0):f2 
   2u 
   4u pl0:f2 
   (center (CA_REB:sp19 ph0):f2 (p7*2 ph0):f3) 
   4u pl0:f2 
   (p10:sp12 ph0):f2 
; ********* CARBONYL END **************   
 
    2u pl0:f2 
 
; ******* CARBON CO-CA CT  *************** 
    (p10:sp17 ph1):f2 
    TAUD_1 pl0:f2 
    (VME_REB:sp9 ph0):f2 
    TAUD_1 pl0:f2 
    2u pl0:f2 
    (CO_REB:sp6 ph0):f2 
    2u pl0:f2 
    (C_REB:sp5 ph0):f2 
    2u pl0:f2 
    TAUD  
    (VME_REB:sp9 ph0):f2 
    TAUD pl0:f2 
    (CO_REB:sp6 ph0):f2 
    2u pl8:f2 
; ******* CARBON CO-CA CT ENDS ************ 
    (p8 ph4):f2 
    4u 
    TAUC 
    4u pl0:f2 
    (CG2_REB:sp15 ph0):f2 

    4u 
    TAUCS 
    4u pl0:f2 
    (C_REB:sp5 ph0):f2 
    4u 
    TAUCS 
    4u pl0:f2 
    (CG2_REB:sp15 ph0):f2 
    4u 
    TAUC_1 
    4u pl0:f2 
  
   (Q5P:sp16 ph14):f2 
    4u 
    TAUC 
    4u pl0:f2 
    pd15    
    4u 
    TAUCS 
    4u pl0:f2 
    (C_REB:sp5 ph0):f2 
    4u 
    TAUCS 
    4u pl0:f2 
    pd15    
    4u 
    TAUC 
    4u pl8:f2 
 
; ******  METHYL EVOLUTION ************* 
  (ralign (p8 ph4):f2 (p1 ph11):f1) 
  2u do:f4 
  d22 
  2u 
  (p1*2 ph0):f1 
  d23 
  2u pl17:f4 
  2u pl0:f2 
  (C_REB:sp5 ph12):f2 
  4u 
  d24 
  (p1*2 ph0):f1 
  d25 
  4u pl8:f2 
; ********* back to protons ********** 

  (p8 ph6):f2 
  2u 
  p25:gp0 
  d4 
  10u fq=cnst22:f2 
  2u pl9:f1 
  (p18 ph9:r):f1 
  2u pl1:f1 
  (center (p8*2 ph0):f2 (p1*2 ph0):f1) 
  2u pl9:f1 
  (p18 ph9:r):f1 
  2u pl1:f1 
  p25:gp0 
  10u pl31:f2 
  d4 
 
  go=2 ph31 cpd2:f2 
  10u do:f2 
  d11 wr #0 if #0 zd 
 
#ifdef CO 
  d12 ip5 
  lo to 3 times 2 
  d12 id0 
  d12 ip31 
  d12 ip31 
  lo to 4 times l2 
  d12 rd0 
#endif 
 
#ifdef METHYL 
  d12 ip6 
  lo to 5 times 2 
  d12 id22 
  d12 id23 
  d12 dd24 
  d12 dd25 
  d12 ip31 
  d12 ip31 
  lo to 6 times l3 
#endif 
 
d11 H2_LOCK 
d11 LOCKH_OFF 
d11 LOCKDEC_OFF 
exit  

ph0=0  
ph1=1 
ph2=0 2 
ph3=1 1 3 3 
ph4=1 1 1 1 3 3 3 3 
ph5=0 0 0 0 2 2 2 2 
ph6=0 
ph7=1 
ph8=2 
ph9=2 
ph10=3 
ph11=1 
ph12=0 
ph14=1 
ph15=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2 
ph31=0 2 0 2 2 0 2 0 0 2 0 2 2 0 2 0 
 


